
3D-L-AN0002: Using Z-Trak with MVTEC HALCON 
HDevelop 

Using Z-Trak with MVTec HALCON HDevelop 
All Z-Trak Laser Profile 3D Sensors.  

Overview 

Z-Trak laser profile 3D sensors are GigE Vision compliant. As such, Z-Trak can be used 

with any image processing software that supports 16-bit image acquisition using the 

GigE Vision protocol. 3D sensor features follow the GenICam standard v3.0 (SFNC 

v2.3). This application note demonstrates how to acquire range images from a Z-Trak 

3D sensor using Halcon HDevelop and display the resulting 3D object model. 

Prerequisites 

The following table lists the recommended Z-Trak firmware and software. 

Z-Trak Firmware Design Software SDK 

2.00 and greater MVTec HALCON  
(18.11 or higher recommended) 

 Sapera LT 8.60 or higher 
(optional, for 3D sensor configuration via Z 

Expert program) 

Software 

MVTec HALCON (full version) machine vision software. Available from the MVTec 

website: 

https://www.mvtec.com/products/halcon/ 

Refer to MVTec documentation for more information. 

The Z-Trak HALCON demo program is available for download from the Teledyne 

DALSA website: 

https://www.teledynedalsa.com/en/products/imaging/3d-sensors/ 

Sapera LT SDK (full version), the image acquisition and control software 

development kit (SDK) for Teledyne DALSA 3D sensors is available for download from 

the Teledyne DALSA website: 

http://teledynedalsa.com/imaging/support/downloads/sdks/ 

Sapera LT includes the Z Expert application, which provides a graphical user interface 

to access 3D sensor features for configuration and setup. 

https://www.teledynedalsa.com/en/products/imaging/3d-sensors/z-trak-3d-laser-profiler/
https://www.mvtec.com/products/halcon/
https://www.teledynedalsa.com/en/products/imaging/3d-sensors/
http://teledynedalsa.com/imaging/support/downloads/sdks/


Doc #: 3D-L-AN0002-V2 October 20, 2020 Page 2 

HALCON Demo Program Overview 
The Z-Trak HALCON demo program demonstrates how to acquire a range image from 

the Z-Trak 3D sensor and convert it to a 3D model ready for processing by Halcon 3D 

point cloud functions. Invalid points are removed from the image to improve image 

display. 

 

Depending on the object being scanned, Z-Trak settings may need to be adjusted 

from default values to obtain high quality profiles and reduce noise. The image height 

should contain enough profiles to image the object. Accurate Y scale is dependent on 

the speed of the object being scanned and the profile rate. 

 

Refer to the Z-Trak user manual for information on how to optimize Z-Trak 

parameters for profile acquisition. 

 

The program performs the following operations: 

 

1. Connects to the Z-Trak and reads features from the device related to image information 
(Width, Height, PixelFormat, AcquisitionLineRate) 

2. Sets the minimum timeout for acquiring a full scan (to avoid a timeout). 
3. Checks the pixel format and grabs a single range map image; the demo program supports the 

Coord3D_C16 pixel format. 
4. Removes invalid pixels from the acquired image using a threshold operation to improve 

visualization. 
5. Obtains scaling factors from the Z-Trak to prepare conversion to 3D model. 
6. Uses the scaling factors retrieved from the 3D sensor to create X, Y, and Z plane images to 

generate a Halcon ObjectModel3D object for 3D visualization and further processing. 

Connecting to the Z-Trak  
The demo program uses the first GigE Vision device found therefore only one Z-Trak 

3D sensor must be connected to the computer running the application. However, if 

necessary, the source code can be modified to select the required device. 

 

The Teledyne DALSA GigE Vision Device Status (installed with Sapera LT) can be used 

to verify if the device is available: 

 

 
 

Alternatively, the HALCON Image Acquisition Assistant, available from the HALCON 

menu bar, can be used to detect devices on the GigEVision2 interface: 

The required device can be selected from the Connection tab from the available 

devices; click Connect to establish communication with the device. 

 

 
 

The Z-Trak feature settings can be modified and are available in the Parameters tab.  

 

 

The profilePerScan feature must be adjusted to acquire more than 1 

profile; set this value to the required number of profiles for a range image. 

Running the Demo Program 
To run the Z-Trak HALCON demo program, load the program and use the Execute > 

Run menu command, click the run icon in the menu bar or use F5. 

 

 
 

 

 

If the Z-Trak 3D sensor is connected to a Halcon application (or other 

application) it must be disconnected before running the demo. 

Upon execution the HALCON Iconic and Control variables are populated in the Variable 

Window. 

The acquired image is rendered as an ObjectModel3D object in the Graphics Window. 

 

 
 

Use the mouse and keyboard combinations to rotate, zoom and move the 3D view. 

 

 
 

Click Continue in the Graphics Window to end the program execution. 

To execute the program again, first use the Execute > Reset Program Execution 

menu command or press F2 before running the program. 

HDevelop Source Code 
 
* Image Acquisition 01: Code generated by Image Acquisition 01 
info_framegrabber ('GigEVision2', 'device', Information, ValueList) 
Device := ValueList[0] 
open_framegrabber ('GigEVision2', 0, 0, 0, 0, 0, 0, 'progressive', -1, 'default', -1, 'false', 
'default', Device, 0, -1, AcqHandle) 
 
* Get image information from device 
get_framegrabber_param (AcqHandle, 'pointsPerProfile', Width) 
get_framegrabber_param (AcqHandle, 'profilesPerScan', Height) 
get_framegrabber_param (AcqHandle, 'PixelFormat', PixelFormat) 
get_framegrabber_param (AcqHandle, 'AcquisitionLineRate', ProfileRate) 
 
* Set minimum timeout for acquiring a full scan (computed in milliseconds with 20% extra) 
MinTimeout := (Height / ProfileRate) * 1000 * 1.2 
set_framegrabber_param (AcqHandle, 'grab_timeout', MinTimeout) 
 
* Check pixel format 
if (PixelFormat == 'Coord3D_C16') 
    * Grab one image from 3D scanner (Height Map) 
    grab_image (HeightMap, AcqHandle) 
else 
    * TODO: add other formats 
endif 
 
* Remove invalid pixels from height map by reducing grayscale domain 
get_framegrabber_param (AcqHandle, 'Scan3dInvalidDataValue', InvalidValue) 
MaxValue := lsh(1,16) - 1 
if (InvalidValue = 0) 
    * Removing 0 from domain 
    MinGray := 1 
    MaxGray := MaxValue - 1 
elseif (InvalidValue = MaxValue) 
    * Removing max value from domain 
    MinGray := 0 
    MaxGray := MaxValue - 2 
else 
    * Invalid value not supported. Skipping domain reduction 
    MinGray := 0 
    MaxGray := MaxValue - 1 
endif 
threshold (HeightMap, Region, MinGray, MaxGray) 
reduce_domain (HeightMap, Region, ImageReduced) 
 
* Get scaling factors for X axis 
get_framegrabber_param (AcqHandle, 'uniformXStepSize', Xscale) 
get_framegrabber_param (AcqHandle, 'aoiFFOVStartX', Xoffset) 
Microns2Meters := 1e-6 
Xscale := Xscale * Microns2Meters 
Xoffset := Xoffset * Microns2Meters 
* Get scaling factors for Z axis 
get_framegrabber_param (AcqHandle, 'heightScaler', Zscale) 
get_framegrabber_param (AcqHandle, 'aoiZStart', Zoffset) 
Zscale := Zscale * Microns2Meters 
Zoffset := Zoffset * Microns2Meters 
* Get scaling factors for Y axis (scaling depends on displacement speed - assuming adjusted for 
square pixels) 
Yscale := Xscale 
Yoffset := 0 
 
* Create Z image with correct scaling in meters 
convert_image_type (ImageReduced, Z, 'real') 
scale_image (Z, Z, Zscale, Zoffset) 
* Create X image with correct scaling in meters 
gen_image_surface_first_order (X, 'real', 0, Xscale, Xoffset, 0, 0, Width, Height) 

* Create Y image with correct scaling in meters 
gen_image_surface_first_order (Y, 'real', Yscale, 0, Yoffset, 0, 0, Width, Height) 
* Generate an ObjectModel3D 
xyz_to_object_model_3d (X, Y, Z, ObjectModel3D) 
 
* Display 3D model 
dev_open_window (0, 0, 800, 600, 'black', WindowID1) 
set_display_font (WindowID1, 16, 'mono', 'true', 'false') 
get_object_model_3d_params (ObjectModel3D, 'num_points', NumPoints) 
if (NumPoints > 0) 
    VisPose := [] 
    Instructions[0] := 'Rotate: Left button' 
    Instructions[1] := 'Zoom:   Shift + left button' 
    Instructions[2] := 'Move:   Ctrl  + left button' 
    visualize_object_model_3d (WindowID1, ObjectModel3D, [], VisPose, ['color_attrib','lut'], 
['coord_z','rainbow'], 'ObjectModel3D', [], Instructions, PoseOut) 
else 
    disp_message (WindowID1, 'No 3D points in the 3D object model!', 'window', 250, 200, 'black', 
'true') 
    stop () 
endif 
     
* Clear the ObjectModel3D 
clear_object_model_3d (ObjectModel3D) 
 
* Close device and exit program 
close_framegrabber (AcqHandle) 


